
News  

US Training Days 
Our US training days are now confirmed for the 15th and 16th of October at the Embassy 
Suites Chicago Downtown hotel, prior to the Society for Neuroscience annual meeting. 
The training days are offered at $250 per day, including continental breakfast buffet and 
lunch as well as coffee and tea, and are suitable for both existing and prospective users 
of Spike2 and Signal. Places are filling up quickly so If you or any of your colleagues 
would like to attend, please follow the links to view the training day program and 
complete the registration form. Please give a short description of the type of research that 
you do and analysis that you are interested in. This will allow us to tailor sessions for 
specific requirements. 

Meetings and events 
Southampton Neuroscience Group (SoNG) Annual Meeting 
University of Southampton 
Southampton, UK 
September 25th 2009 

Neuroscience 2009 - 39th Annual Meeting 
McCormick Place 
Chicago, IL, USA 
October 17th – 21st 2009 

8th Biennial conference of the Chinese Society for Neuroscience (CSfN) 
Guangzhou Huatai Hotel 
Guangzhou, China 
November 7th – 10th 2009 

Federation of European Physiological Societies 2009 
University Medical Centre Ljubljana 
Ljubljana, Slovenia 
November 12th – 15th 2009 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 7.01 Signal version 4.05 

Spike2 version 6.12 Signal version 3.12 

Spike2 demo Signal demo 

Spike2   

Q. Is there any way to create data files larger than 2GB when sampling?  

A. Spike2 now supports data file sizes of up to 1TB when using the Big file format 
option in the sampling configuration Channels tab. 

There are two limits to the size of a Spike2 data file: the number of clock ticks in 
the file (limited to 2147483647) and the physical size of the file. The physical size 
before version 7 was limited to 2147483647 bytes, or 2 GB. For most users of 
Spike2, these limits are not a problem. However, it is possible to run out of disk 
space before you run out of clock ticks. For example, if you set the Waveform 
sample rate to be one clock tick and sample one channel, each sample point 
uses 2 bytes of data. If you set Burst mode sampling in the Resolution tab with n 
channels, each taking one sample per clock tick, then each clock tick is using 2n 
data bytes. This means that your maximum possible run time is limited to 
2147483647/2n clock ticks. 

 
Sampling configuration with Big file option enabled 

  Without using Big file mode, if you were sampling 32 channels of data at 31.25 
kHz each (the maximum possible with a Power1401 mk II), in non-burst mode 
you would need to set the clock tick to 1 microsecond, which means that the 
clock tick limit would be some 35 minutes of sampling. However, the file size limit 
would be hit in around 17 minutes. If you changed to burst mode sampling, the 
clock could be run at 32 microseconds per tick, which extends the clock limit to 
19 hours, but the file size limit is still 17 minutes.  

If you check the Big file option, the maximum file size is increased by a factor of 
512 allowing sampling for the full time allowed by the clock tick period.  

 

 

Back to top

 

CED
You can create data files with space for up to 400 data channels using the Set... button in the Channels tab of the sampling configuration



  

  

 

Q. Is Signal suitable for analysis of single channel patch clamp data? 

A. There are a number of features for single channel patch clamp data analysis 
available, including idealized trace formation and generation of dwell time and 
amplitude histograms. An idealized trace is formed by converting the raw data 
trace into a series of open/closed times (events). Each time has an amplitude, a 
duration and a set of ‘flags’ to identify it.  

 There are two methods of creating a new idealised trace from the Analysis menu 
Open/Closed times command: Threshold and SCAN. The Threshold method 
involves setting an Open and Closed threshold level to detect the opening of ion 
channels for inward and outward currents. This method is more suitable if you 
have more than one channel in your patch and the resultant trace is drawn 
overlaying the raw data. The SCAN analysis method (originally developed and 
written by professor David Colquhoun at University College London) assumes 
that a Gaussian filter was used to remove noise from the original signal and 
generates a high time resolution calculation of the original, unfiltered waveform 
using a form of reverse convolution of the idealized trace with the step response 
function of the filter. The result is drawn with the convoluted trace overlaying the 
data and the idealized trace drawn below. 

 

  

 
Idealised traces generated using the threshold method (top) and SCAN analysis 

method (bottom) 

 Once an idealised trace has been generated from the data, you can view and 
edit details of the events by right-clicking on the trace and selecting Show event 
details from the Edit menu pull-down. You can also click and drag the selected 
level on the trace to adjust amplitude and dwell times. Open/Closed time 
histograms, Amplitude histograms of the open/closed time durations and burst 
histograms can all be created from the Analysis menu. The settings for all 
histograms allow you to include or exclude data based on the associated ‘flags’. 

Settings for sampling of patch and voltage clamp data are available in the Clamp tab of the sampling configuration dialog if these options were selected during installation



 Further details on idealized trace generation and analysis can be found in the on-
line help. 

Scripts: Spike2  
Q. I would like to be able to take channels from multiple waveform averages and 

merge them into a single file so that I can display them overlayed. Is there a 
script available that could do this? 

A. The attached script, MergeODResultViews.s2s, creates a new XY window 
displaying overdrawn data from multiple result views selected in turn by the user. 
By default the script takes data from the first channel of the result view and 
displays each channel (up to a maximum of 256) overdrawn and in separate 
colours in the new window. 

Scripts: Signal  
Q. I would like to be able to trigger sampling in Signal when a designated channel is 

between two threshold levels. Peri-trigger mode allows me to trigger when there 
is a transition from within to outside a pair of levels, but not when the waveform is 
between levels. Is there any way to achieve this? 

A. The attached script, TrigInLevels.sgs, and sequence file, 
TrigInLevels.pls, waits until a sampled waveform is between two levels on 
ADC port 0 (channel 1) before triggering sampling and then waits for a defined 
time before arming the next sweep. The threshold levels are defined using 
horizontal cursors and the script uses peri-triggered sampling mode and free run 
without restarts mode for the outputs. This sequence could easily be modified to 
generate additional outputs for the sweep or respond to multiple states conditions 
if required. 

Did you know…? 
You can now update the Spike2 output sequencer by loading sequence files during 
sampling. The new sequencer file replaces the previous one. 

Recent questions   
Q. Are there keyboard shortcuts in Spike2 for optimising the data in all channels in a 

file? This would be much easier than opening the Y-axis dialog and setting all 
channels to optimise. 

A. There are many functions in both Spike2 and Signal that have associated 
 ‘hotkey’ keyboard shortcuts. The Ctrl+Q shortcut optimises all channels in a file. 

 A full list of keyboard shortcuts can be found in the on-line help for both Spike2 
and Signal. 

Customer contribution – 4 channel dynamic clamp implementation using Spike2 
and 1401plus 
Dynamic clamp methods are a standard in electrophysiology that increase flexibility in the 
manipulation of membrane processes—e.g., addition and subtraction of ion channels, 
synapses, and additional component neurons in a network—however, these methods’ 
time resolution and stability are limited by computing speed and model complexity.  In 
this application, we used the built-in real time output capability of Spike2 with a 1401plus 
interface to implement a dynamic clamp to arbitrarily rewire a motor pattern generator in 
vitro.   

Using up to 4 intracellular recordings from different neurons, we added and subtracted 
electrical couplings between these neurons.  This was done using a sequencer file 
containing a loop that calculated currents to inject into each neuron at precise, small 
timesteps, i.e. in “real time”. These currents were solutions to Ohm’s Law between 
neurons and Kirchoff’s Current Law at each node of the network (where each neuron was 
treated as a node).  The attached script (4ChanDynClamp.s2s) and sequence file 
(4NeuronCoupling.pls) is a basic example of implementing a 4 channel dynamic 
clamp using user-defined settings that define the connectivity of the network (coupling of 
neurons) and a global conductance value, G, that defines the strength of every 
connection. Editing of these settings and sampling of data is controlled via the buttons in 
the script toolbar. 

'$MergeResultViews|Forms a new window showing overdrawn channels from multiple result views. 
'You must make the result view to 'ADD' the current view.
'Not checking for errors in terms of different sized arrays or whether error bars are available in results.

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 09/09/09
 
Var result%;            'View handle for XY result
Var Avg%;              'View handle of current result view
Var Bins;                'bin increments
Var ok%;                'Interact result
var data[10000];        'array for data points
var c%:=1;              'Current channel in XY view
var noBins;              'No of bins (data points)
var sweep;              'Length of result view

Avg%:=FrontView();    'Get current selection

If ViewKind(avg%)<>4 then
   message("Please make the average the front view");    'Check that it's a result view and warn user if not
Halt
endif;

bins:=binsize();                'Get bin increment
sweep:= BinSize()*Maxtime();         'Get length
noBins:=XHigh()-Xlow();               'Get no of bins (data points)

ArrConst(data[],view(avg%).[]);        'Copy data from result view

result%:= FileNew(12);       'Create XY result window
DrawMode(1,2,0,1);           'Draw as lines
View(result%).XYAddData(c%,data[:noBins],bins);     'Add data to first channel of result
XYJoin(c%,1);                  'Force joined points
XYColour(c%,c%);              'Set colour from palette
XYDrawMode(c%,2,0);         'Set point size
XRange(0.0,sweep);            'Show length of result
Optimise(-1);                   'Optimise
WindowTitle$("Merged and overdrawn result view");  'Title window 

repeat                 'Repeat for selected result views
ok%:=Interact("Select another result view",2+4+8+16+32+64+128+256+512,0,"Quit","Add");
    If ok%=1 then halt endif            'if Quit then halt script
    ArrConst(data[],0.0);          'Initialise data array
    XYSetChan(0);                'Set a new channel in the XY result
    c%+=1;                       'Increment channel no.
    Avg%:=FrontView();           'Get current window
    FrontView(avg%);               'Set to front
    If ViewKind(avg%)<>4 then
    Message("Please make the result view the front view");    'Check if result view
    Halt
    endif;
    bins:=binsize();              'Get bin increment
    sweep:= BinSize()*Maxtime();      'Get sweep length
    noBins:=XHigh()-Xlow();            'Get no. of bins (data points)
    ArrConst(data[],view(avg%).[]);    'Copy result view data
    View(result%);                       'Make sure XY result is current
    XYAddData(c%,data[:noBins],bins);      'Add result view data
    XYJoin(c%,1);                 'Set joined data points
    XYColour(c%,c%);             'Set new colour
    XYDrawMode(c%,2,0);         'Set point size
    
Optimise(-1);          'Optimise all channels
Until ok%=1;

FrontView(result%);       'Bring XY result to front


CED
MergeODResultViews.s2s


'$TrigInLevels|Script to trigger sampling when the input signal on ADC port 0 lies within two trigger levels defined by horizontal cursors
'Must be used with the associated sequence file, TrigInLevels.pls

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 14/09/09

var data%;          'View handle for data file
var inputChan%:= 1;     'Input channel to monitor = ADC port 0
var oldLoTrigLev,oldHiTrigLev;      'Previous trigger levels
var loTrigLev,hiTrigLev;        'Current trigger levels
var loTrigVar,hiTrigVar;        'Current trigger levels in ADC units

SampleMode(1);      'Set peri-trigger mode
SamplePeriType(4);      'Set event for peri-trigger type
SampleSeqCtrl(2,0);     'Set free run without restarts
New%();         'Create a new data file ready to sample
DoToolbar();        'Control via toolbar

Func New%();
data%:=FileNew(0,2);			'Open a new data file for sampling
HCursorDelete(-1);           'Clear existing horizontal cursors  
WindowVisible(1);           'Make sure window is visible
if data%<0 then Message("Unable to open new data file");Halt() endif;
Window(0,0,100,100);						'Make data window full screen
HCursorNew(inputChan%,oldLoTrigLev);				'Set cursors on appropriate channel
HCursorNew(inputChan%,oldHiTrigLev);				
HCursorLabel(4,1,"Lower: %p");
HCursorLabel(4,2,"Upper: %p");
Interact("Position cursors at initial threshold levels",1023);      'User positions cursors
oldLoTrigLev:= HCursor(1);					'Get threshold levels
oldHiTrigLev:= HCursor(2);
loTrigVar:= (HCursor(1) - SamplePortZero(0)) * 32768 / (SamplePortFull(0)-SamplePortZero(0)); 'Convert to ADC units
hiTrigVar:= (HCursor(2) - SamplePortZero(0)) * 32768 / (SamplePortFull(0)-SamplePortZero(0));
SampleSeqVar(1,loTrigVar);				'Set output sequencer variables
SampleSeqVar(2,hiTrigVar);

FrontView(data%);						'Make sure the data view is at the front
ToolbarEnable(3,0);						'Disable "Sample stop" button
ToolbarEnable(2,1);						'Enable "Sample start" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

proc DoToolbar()
ToolbarSet(1,"Quit",Quit%);				'Set up toolbar buttons
ToolbarSet(2,"Sample start", Start%);
ToolbarSet(3,"Sample stop", Stop%);
ToolbarSet(4,"New file", New%);
ToolbarEnable(3,0);						'Disable "Sample stop" button
View(data%);
Toolbar("Press SAMPLE START to commence sampling", 1023);					'Wait here until quit is pressed
end;

func Quit%()							'If "Quit" is pressed
SampleStop();							'Stop sampling
return 0;								'leave toolbar
end;

func Start%()							'If "Start" is pressed
ToolbarSet(0,"",Idle%);					'Call Idle%() whenever there is free time
SampleStart();							'Start sampling
ToolbarEnable(4,0);						'Disable "New file" button
ToolbarEnable(3,1);						'Enable "Sample stop" button
ToolbarEnable(2,0);						'Disable "Sample start" button
ToolbarEnable(1,0);						'Disable "Quit" button
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;								'Stay with toolbar
end;

func Stop%()							'If "Stop" is pressed
ToolbarClear(0);
SampleStop();							'Stop sampling
ToolbarEnable(4,1);						'Enable "New file" button
ToolbarEnable(3,0);						'Disable "Sample stop" button
ToolbarEnable(1,1);						'Enable "Quit" button
ToolbarText("Press FILE NEW to capture more data");
return 1;								'Stay in toolbar
end;

func Idle%()							'The Idle routine is called when PC has time
View(data%);

if SampleStatus() = 2 then
    loTrigLev:=HCursor(1);								'Check if threshold levels have changed
    hiTrigLev:= HCursor(2);
    
    if loTrigLev <> oldLoTrigLev or hitrigLev <> oldHiTrigLev then					'If so, Get new levels in ADC units
       loTrigVar:= (loTrigLev - SamplePortZero(0)) * 32768 / (SamplePortFull(0)-SamplePortZero(0)); 'Convert to ADC units
       hiTrigVar:= (hiTrigLev - SamplePortZero(0)) * 32768 / (SamplePortFull(0)-SamplePortZero(0)); 

        SampleSeqVar(1,loTrigVar);						'Update sequencer variables		
        SampleSeqVar(2,hiTrigVar);
        oldLoTrigLev:= loTrigLev;							'Reset previous cursor level to current level
        oldHiTrigLev:= hiTrigLev;
    endif;
endif;
return 1;								'Stay in toolbar
end;


CED

'$4ChanDynClamp|Example script to implement a basic 4 channel dynamic clamp setup with
'the associated sequence file 4NeuronCoupling.pls

'Script to control sequencer parameters while running.
'Note that input channels MUST be = [5,6,7,8] and DAC output channels MUST be = [0,1,2,3],
'output channel 1 is for neuron on channel 5
'output channel 2 is for neuron on channel 6
'output channel 3 is for neuron on channel 7
'output channel 0 is for neuron on channel 8

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.


'Fred Sieling & CED 11/09/2009

var data%;               'View handle of new data file
var nneurons%:=4;       'No. of neurons (channels) recorded
var nconn%:=6;           'No. of possible connections between neurons
var indconselect%[6];    'Array for coupling selection
var chkbxmem%[6];       'Temp. array for last settings
var n%;
var tempind%;            'Temp. index for settings
var newG;                 'New Global value for calculations
var x%[6],y%[6];                'Arrays to hold channel no.s for coupling selection
var txtchan%;             'Flag to use TextMark
var GS;                    'Global value for sequencer

for n%:=0 to 5 do
    indconselect%[n%]:=n%;
next;
x%[0]:=5; x%[1]:=6; x%[2]:=7; x%[3]:=8; x%[4]:=5; x%[5]:=6;           'Setup channel array for coupling selection
y%[0]:=6; y%[1]:=7; y%[2]:=8; y%[3]:=5; y%[4]:=7; y%[5]:=8;

HideAll();                      'Hide all unwanted bars
ToolbarVisible(1);              'Make toolbar visible always
New%();                         'Set up new sampling window
DoToolbar();                    'Do the toolbar
RestoreAll();                   'Restore all bars
Halt;


func New%()                     'New sampling window
View(LogHandle());              'Make log view the current view
EditSelectAll();                'Select all text in log view
EditClear();                    'Delete it
Window(0,80,100,100);           'Display it at the bottom of the screen
WindowVisible(1);               'Make it visible

if data%>0 then                 'If there is already a data view open...
  View(data%);                  '...close it
  FileClose();                  'No point setting data% to 0...
endif;

data%:=FileNew(0,1);            'Open a new data file for sampling
if data%<0 then Message("Unable to open new data file");Halt() endif;
DrawMode(-1,2);                 'Set event draw mode to lines
Window(0,0,100,80);             'Make data window in top bit of screen
XRange(0,10);

FrontView(LogHandle());         'Bring the Log view to the front
FrontView(data%);               'Bring the data view to the front
ToolbarEnable(3,0);             'Disable "Sample stop" button
ToolbarEnable(2,1);             'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

proc DoToolbar()
ToolbarSet(1,"&Quit",Quit%);    'Set up toolbar buttons
ToolbarSet(2,"&Sample start", Start%);
ToolbarSet(3,"&Sample stop", Stop%);
ToolbarSet(5,"Edit Sequence", edSequence%);
ToolbarSet(7,"&New file", New%);
ToolbarEnable(3,0);             'Disable "Sample stop" button
ToolbarEnable(5,0);             'Disable "Edit Sequence" button
Toolbar("Press SAMPLE START to commence sampling", 1023);'Wait here until quit
end;

func Quit%()                    'If "Quit" is pressed
SampleStop();                   'Stop sampling
return 0;                       'leave toolbar
end;

func Start%()                   'If "Start" is pressed
SampleStart();                  'Start sampling
ToolbarEnable(4,0);             'Disable "New file" button
ToolbarEnable(3,1);             'Enable "Sample stop" button
ToolbarEnable(5,1);              'Enable "Edit Sequence" button
ToolbarEnable(2,0);             'Disable "Sample start" button
ToolbarEnable(1,0);             'Disable "Quit" button
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;                       'Stay with toolbar
end;

func Stop%()                    'If "Stop" is pressed
SampleStop();                   'Stop sampling
if (data%>0) and                'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);             'Enable "New file" button
ToolbarEnable(3,0);             'Disable "Sample stop" button
ToolbarEnable(1,1);             'Enable "Quit" button
ToolbarText("Press FILE NEW to capture more data");
return 1;                       'Stay in toolbar
end;

Func edsequence%()
'**********************************************************
'Dialog Box 
var i%,ok%;
var couplenow%[nconn%];     
for n%:=0 to nconn%-1 do    'put old settings into temp var
    tempind%:=indconselect%[n%];
    couplenow%[n%]:=chkbxmem%[tempind%];
next;

DlgCreate("Edit Sequencer");
DlgReal(1,"Global Value G (nS)",0,200);
for i%:=2 to 2+nconn%-1 do
    tempind%:=indconselect%[i%-2];           'Get and display each channel coupling combination
    DlgCheck(i%,Print$("Couple Channel %d to %d",x%[tempind%],y%[tempind%]));
next;
DlgCheck(1+nconn%+1,"Output to TextMark");
ok%:=DlgShow(newG,couplenow%,txtchan%);
if ok%=0 then
    return 1
endif

for n%:=0 to nconn%-1 do    'put new settings into globalvar for next update
    tempind%:=indconselect%[n%];
    chkbxmem%[tempind%]:=couplenow%[n%];
next;
'**********************************************************
'Write to Sequencer
GS:=65536*newG/1000*10;  'value converted to 32-bit integer for sequencer

For n%:=0 to nconn%-1 do
    SampleSeqVar (n%+6,couplenow%[n%]*GS);    'set new Gxy values.. zero if turned off in dialog box
    PrintLog("Set SeqVar %d to value %d (nS)\n",n%+6,couplenow%[n%]*newG);
Next;
'**********************************************************
'Write to textmark channel
if txtchan% then 'if nonzero channel entry for textmark channel in dialog
    var txtstr$;
    txtstr$:=Print$("G=%4.1fnS;Couplage A:B=(1/0);",newG);   '28 chars
    for n%:=0 to nconn%-1 do
        tempind%:=indconselect%[n%];
        txtstr$+=Print$("%d:%d=%d;",x%[tempind%],y%[tempind%],couplenow%[n%]);   '7 chars each * 6 groups = 42 chars
    next;
        PrintLog(txtstr$+"\n"); 
        SampleText(txtstr$);
endif
return 1
end;

'You can find this code in the App() on-line help.
var gFloat%[20];                'global for floating window states
proc HideAll()
var i%;
gFloat%[0] := App(-2);          'number of windows
for i% := 1 to gFloat%[0] do    'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do    'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end

The 1401plus sequencer has a minimum sequencer clock tick of 50 µs, meaning that it 
can complete one instruction in this time, e.g., addition of two variables or reading a 
voltage from an analog input, so efficient coding and our choice of linear electrical 
couplings rather than nonlinear synaptic couplings was important- the tradeoff is between 
model complexity and stability of the dynamic clamp.  Briefly, large oscillations on the 
electrode occur when the sequencer loop runs too slowly.  Our most useful sequencer 
files had ~20 instructions and completed one loop every ~1 ms, reading voltages from 
each neuron, calculating currents to inject, and writing command voltages to be sent to 
the current clamp amplifier via the 1401 DAC outputs.   

While 1 ms is slow compared to nominal update intervals of 50 to 100 µs used in 
contemporary dynamic clamp experiments, we found that 1 ms was sufficiently fast to 
show a nice effect in our experiments while maintaining stability.  This was possible partly 
because we injected relatively small magnitudes of current and balanced the bridge often. 
A modern Power1401 interface, which has a minimum sequencer step time of 10 µs 
would allow speeding up of sequencer execution by a significant factor. 

Fred Sieling 
Laboratory for Neuroengineering 
Georgia Institute of Technology 

User group      
Try the new CED Forums bulletin board for software and hardware support 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

